'****************************************************************************

'*                      GESTIONE RICEVITORE GPS                             *

'*    IL PROGRAMMA SUPPORTA IL PROTOCOLLO NMEA 0183 VERSIONE 2.0            *

'* -------------------------------------------------------------------------*

'*                                                                          *

'* Implementate solo funzionalità base. Viene decodificato il messaggio     *

'* GPRMC contenente dati essenziali.                                        *

'* Vedi articolo sui moduli GPS su ELETTRONICA IN No 28 (Aprile 98)         *

'*                                                   IK3SSQ,  Damir Silic   *

'****************************************************************************

' Il programma è composto da questo modulo principale più alcune subroutines.

' In più è definita una funzione SCAN che serve per trasformare i dati

' in arrivo in una forma più leggibile, e più facilmente gestibile da altri

' programmi (data base, fogli elettronici ecc.) il formato ottenuto è il

' seguente:

'    "GPRMC;145835;A;4533.0713;N;009000.9995;E;5.1;257.0;040398;000.0;W"

' I campi sono separati da ;

' Le subroutines sono

' RICEZ     routine di ricezione dati da porta seriale

' CHECKSUM  routine di calcolo del checksum secondo specifiche NMEA 0183

' GESTIONE  routine di gestione dei messaggi (solo GPRMC per ora)

' SAVE      routine di salvataggio dati su file

'--------------------------------------------------------------------------

'***************** INIZIO MODULO PRINCIPALE *******************************

' Dichiarazione delle subroutine del programma

DECLARE SUB save (g%)

DECLARE SUB checksum (msg$)

DECLARE SUB gestione ()

DECLARE SUB ricez ()

DECLARE FUNCTION SCAN% (A$)

' Definisce variabili intere dalla A alla Z

DEFINT A-Z

' Definisce variabili che devono essere condivise tra i vari moduli del prog.

COMMON SHARED ck, ckl, ckh, inizio, fine, ric$, stx$, etx$, com$

COMMON SHARED g$(), SALVA

' Preparazione schermo (grafica elementare)

CLS

SCREEN 9

PRINT TAB(29); "Global Positioning System "

LOCATE 2, 2: PRINT "NMEA 0183 V 2.0"

BEEP

CIRCLE (320, 91), 30, 9

PAINT (320, 91), 9

CIRCLE (250, 90), 6, 12

PAINT (250, 90), 12

CIRCLE (390, 90), 6, 12

PAINT (390, 90), 12

CIRCLE (320, 40), 6, 12

PAINT (320, 40), 12

CIRCLE (320, 141), 6, 12

PAINT (320, 141), 12

CIRCLE (320, 90), 70, 12

LOCATE 23, 20: PRINT "ESC per uscire-F1 salvataggio ON/OFF"

' Dimensionamento stringa di ricezione

DIM g$(50)

' Preparazione riconoscimento tasti (F1-F10)

F1$ = CHR$(0) + ";"

f2$ = CHR$(0) + "<"

f3$ = CHR$(0) + "="

f4$ = CHR$(0) + ">"

f5$ = CHR$(0) + "?"

f6$ = CHR$(0) + "@"

f7$ = CHR$(0) + "A"

f8$ = CHR$(0) + "B"

f9$ = CHR$(0) + "C"

f10$ = CHR$(0) + "D"

' Carattere inizio trasmissione

stx$ = CHR$(36)

' Carattere fine trasmissione

etx$ = CHR$(10)

' Imposta la variabile salva=1  salvataggio su file abilitato

SALVA = 1

' Inizializza la porta seriale  disabilitando i segnali di controllo '

' (si usano solo tre fili)                                           '

' La porta seriale viene definita come device immediato 1 (#1)

' Random sta per bidirezionale

' Dichiara i parametri fissi per la com

par$ = ",rs,cd0,ds0,cs0"

baud$ = "4800,n,8,1"

' Richiede l' input della porta desiderata

LOCATE 15, 30: INPUT "Quale porta com usi "; c$

' Cancella la scritta dallo schermo

LOCATE 15, 30: PRINT "                         "

' Costruisce la stringa di inizializzazione con parametri appropriati

com$ = "com" + c$ + ":"

set$ = com$ + baud$ + par$

' Configura la porta com

OPEN set$ FOR RANDOM AS #1

' Il comando dato ha il seguente aspetto:

'OPEN "com2:4800,n,8,1,rs,cd0,ds0,cs0" FOR RANDOM AS #1

' Quando viene ricevuto un carattere sulla com salta sulla routine ricevi

ON COM(VAL(c$)) GOSUB ricevi

' Abilita la porta

COM(VAL(c$)) ON

LOCATE 22, 42: PRINT "Wait for receive GPS data on "; com$

' Attiva la funzione di controllo tastiera

DO

' Gestione tasto ESC per uscire dal programma

A$ = INKEY$

SELECT CASE A$

CASE CHR$(27)

COM(VAL(c$)) OFF

CLOSE #1

BEEP

END

' Gestione tasto F1 per attivare/disattivare salvataggio su file

CASE F1$

IF SALVA = 1 THEN

SALVA = 0

ELSE

SALVA = 1

END IF

' Prova attivazione tasto F2

CASE f2$

LOCATE 9, 1: PRINT "PROVA PRESSIONE TASTO F2"; SALVA

CASE ELSE

' Altri tasti non sono gestiti

END SELECT

LOOP

ricevi:

ricez

RETURN

'FINE MODULO PRINCIPALE

'**************************************************************************

SUB checksum (msg$)

' Procedura per il calcolo del checksum secondo le specifiche NMEA0183

ck = 0

FOR k = 1 TO LEN(msg$)

ck = ck XOR ASC(MID$(msg$, k, 1))

NEXT k

' La variabile ck contiene il checksum calcolato

END SUB

SUB gestione

' Procedura di gestione dei diversi messaggi del protocollo

' Estrae dalla stringa ricevuta l' informazione sul tipo del messaggio

com$ = MID$(ric$, 1, 5)

SELECT CASE com$

' Il messaggio Š un GPRMC ? (viene gestito solo questo)

CASE "GPRMC"

' Estrapolazione e stampa dati essenziali del messaggio GPRMC

g = SCAN(ric$)

LOCATE 11, 52: PRINT "Record ricevuto=" + g$(0)

ore$ = MID$(g$(1), 1, 2)

min$ = MID$(g$(1), 3, 2)

sec$ = MID$(g$(1), 5, 2)

ora$ = ore$ + ":" + min$ + ":" + sec$

giorno$ = MID$(g$(9), 1, 2)

mese$ = MID$(g$(9), 3, 2)

anno$ = MID$(g$(9), 5, 2)

data$ = giorno$ + "/" + mese$ + "/" + anno$

LOCATE 10, 1: PRINT "Data="; data$; ",Ora UTC="; ora$

LOCATE 11, 1: PRINT "Stato del GPS=";

IF g$(2) = "A" THEN

LOCATE 11, 15: PRINT "OK    "

LOCATE 12, 1: PRINT "Latitudine="; g$(3); "-"; g$(4)

LOCATE 13, 1: PRINT "Longitudine="; g$(5); "-"; g$(6)

LOCATE 14, 1: PRINT "Velocit…="; g$(7)

LOCATE 15, 1: PRINT "Direzione di movimento="; g$(8)

LOCATE 16, 1: PRINT "Variazione magnetica="; g$(10)

LOCATE 17, 1: PRINT "Direzione variazione magnetica="; g$(11)

' Salvataggio dati su file (se attivo)

IF SALVA = 1 THEN

LOCATE 10, 52: PRINT "SALVATAGGIO DATI ATTIVO    "

save (g)

ELSE

LOCATE 10, 52: PRINT "SALVATAGGIO DATI NON ATTIVO"

END IF

ELSE

LOCATE 11, 15: PRINT "ERRORE"

END IF

' PRINT

CASE ELSE

' Altri messaggi non sono supportati

END SELECT

END SUB

SUB ricez

'Procedura di ricezione dati secondo protocollo NMEA 0183

LOCATE 22, 42: PRINT "Receive data at 4800 8N1 in progress"

' Pone la condizione di fine trasmissione

' Finchè End Of File 1 (device aperto è la porta seriale) non si verifica

' Resta in questa routine

WHILE NOT EOF(1)

' Legge 1 carattere dal device aperto

char$ = INPUT$(1, 1)

' Scrive il carattere appena letto (giusto per vedere che qualcosa arriva)

LOCATE 21, 42: PRINT "Port activity-->"

LOCATE 21, 59: PRINT LEFT$(char$, 1)

' Analisi del carattere ricevuto

SELECT CASE char$

' Se il carattere è stx (marker di inizio messaggio)

CASE stx$

' Imposta parametri di inizio messaggio

inizio = 1

fine = 0

ric$ = ""

' Se il carattere è etx (marker di fine messaggio)

CASE etx$

' Imposta parametri di fine messaggio (solo se stava già arrivando)

IF inizio = 1 THEN

fine = 1

' Prepara la striga per l' elaborazione (toglie il carattere stx, etx

' estrae il checksum ricevuto (ckr$) e poi lo toglie dalla stringa ricevuta

ric$ = LEFT$(ric$, LEN(ric$) - 1)

ric$ = RIGHT$(ric$, LEN(ric$) - 1)

ckr$ = MID$(ric$, LEN(ric$) - 1, 2)

ric$ = LEFT$(ric$, LEN(ric$) - 3)

END IF

END SELECT

' Fine analisi del carattere ricevuto

' Controlla la progressione del messaggio

IF inizio = 1 THEN

IF fine = 0 THEN

' Se entrambe vere il messaggio non è finito e salvo il carattere ricevuto

ric$ = ric$ + char$

' Controlla l'overflow del messaggio

IF LEN(ric$) > 220 THEN

' Non ho ricevuto etx dopo 220 caratteri

' Reseta i parametri e anulla la stringa ricevuta

inizio = 0

fine = 0

ric$ = ""

END IF

' Altrimenti (fine=1) il messaggio è stato ricevuto

ELSE

' Finita tutta la ricezione compreso il checksum

' Vado a calcolare il checksum della stringa ricevuta

checksum (ric$)

' Se il checksum ricevuto è diverso da quello calcolato va in errore

IF RIGHT$("00" + HEX$(ck), 2) <> ckr$ THEN

LOCATE 11, 50: PRINT "-->Errore nel trasferimento seriale"

' Altrimenti chiama la routine di gestione

ELSE

gestione

END IF

' Se siamo in questo punto

' Il messaggio ricevuto è coretto ed è stato trattato dalla routine di

' gestione. Oppure si è verificato un errore di checksum

' In ogni caso:

' Imposta i parametri per ricezione nuovo messaggio

inizio = 0

fine = 0

END IF

' Se siamo in questo punto:

' Il messaggio non è terminato oppure non è ancora iniziato

' oppure si è verificato un errore di overflow

END IF

' Ricomincia dall' inizio finchè la ricezione non è terminata(WHILE NOT EOF)

WEND

END SUB

'

SUB save (g)

' Procedura di salvataggio dati su un file ASCII

' Apertura file di salvataggio

OPEN "FILEOUT.TXT" FOR APPEND AS #2

FOR T = 0 TO g - 2

PRINT #2, g$(T); ";";

NEXT T

CLOSE #2

END SUB

'

FUNCTION SCAN (A$)

' Procedura per la suddivisione automatica del messaggio in arrivo

g = 0

A = 1

aold = 1

DO

A = INSTR(A, A$, ",")

IF A > 0 THEN

A = A + 1: x$ = MID$(A$, A, 1)

r$ = MID$(A$, aold, (A - aold) - 1)

IF r$ = "" THEN

g$(g) = "NULL"

ELSE

g$(g) = r$

END IF

aold = A

ELSE

A = LEN(A$) + 1

r$ = MID$(A$, aold, A - aold)

g$(g) = r$

EXIT DO

END IF

g = g + 1

LOOP

SCAN = g + 1

END FUNCTION

